第六章 µC/OS-II在STR71x微控制器上的移植

6.1 µC/OS-II简介
µC/OS-II是一个完整的、可移植、可固化、可裁剪的抢占式实时多任务内核。µC/OS-II是用ANSI的C语言编写的，包括一小部分汇编语言代码，使之可供不同架构的微处理器使用。至今，从8位到64位，µC/OS-II已在超过40种不同架构的微处理器上运行。不过，µC/OS-II并不是免费使用的，有关版权、版税等问题，可联系µC/OS-II的代理商。

µC/OS-II特点

源码公开 购买《嵌入式实时操作系统µC/OS-II（第2板）》可以获得µC/OS-II V2.52版本的所有源代码。购买此书的其它版本可以获得相应版本的全部源代码。

可移植性（portable） µC/OS-II的源代码绝大部分是用移植性很强的ANSI C写的，与微处理器硬件相关的部分是用汇编语言写的。汇编语言写的部分已经压到最低限度，以使µC/OS-II便于移植到其它微处理器上。µC/OS-II可以移植到许多微处理器上，条件是：该处理器具有堆栈指针，具有CPU内部寄存器入栈、出栈指令。另外，使用的C编译器必须支持内嵌汇编（in line assembly），或者C语言可扩展和可链接汇编模块，使得关中断和开中断能在C语言程序中实现。µC/OS-II可以在绝大多数8位、16位、32位以及64位微处理器、微控制器及数字信号处理器（DSP）上运行。

可固化（ROMable） µC/OS-II是为嵌入式应用而设计的，这就意味着，只要具备合适的系列软件工具（C编译、汇编、链接及下载/固化），就可以将µC/OS-II嵌入到产品中作为产品的一部分。

可裁剪（scalable） 可以只使用µC/OS-II中应用程序需要的系统服务。也就是说，某个产品可以只使用很少几个µC/OS-II调用，而另一个产品则可能使用了几乎所有µC/OS-II的功能，这样可以减少产品中µC/OS-II所需的存储空间（RAM和ROM）。µC/OS-II使用条件编译实现可裁减，只要在用户的应用程序中（用#define constants语句）定义那些需要的µC/OS-II功能就可以了。

可剥夺性（preemptive） µC/OS-II是完全可剥夺型的实时内核，即µC/OS-II总是运行就绪条件下优先级最高的任务。大多少商业内核也是可剥夺型的，µC/OS-II在性能上与它们类似。

多任务 µC/OS-II可以管理64个任务；然而，µC/OS-II作者建议用户保留8个给µC/OS-II。留给用户的应用程序最多可有56个任务。赋予每个任务的优先级必须是互不相同的，这意味着µC/OS-II不支持时间片轮转调度法（round-robin scheduling）。该调度法适用于调度优先级相同的任务。

可确定性 绝大多数µC/OS-II的函数调用和服务的执行时间具有可确定性。也就是说，用户总是能知道µC/OS-II的函数调用与服务执行了多长时间。进而可以说，除了函数OSTimeTick()和某些事件标志服务，µC/OS-II系统服务的执行时间不依赖于用户应用程序任务数目的多少。

任务栈 每个任务都有自己单独的栈。µC/OS-II允许每个任务有不同的栈空间，以便压低应用程序对RAM的需求。使用µC/OS-II的栈空间校验函数，可以确定每个任务到底需要多少栈空间。

系统服务 µC/OS-II提供很多系统服务，例如信号量、互斥型信号量、事件标志、消息邮箱、消息队列、信号量、块大小固定的内存的申请与释放及时间管理函数等。

中断管理 中断可以使正在执行的任务暂时挂起。如果优先级更高的任务被中断唤醒，则高优先级的任务在中断嵌套全部退出后立即执行，中断嵌套层数可达255层。

稳定性与可靠性 µC/OS-II是基于µC/OS的。µC/OS自1992年以来已经有数百个商业应用。µC/OS-II与µC/OS的内核是一样的，只是提供了更多的功能。另外，2000年7月，µC/OS-II在一个航空项目中得到了美国联邦航空管理局对用于商用飞机的、符合RTCA DO-178B标准的认证，该标准对用于航空设备方面的软件提出了要求。为了符合这一标准，必须尽可能地通过文件描述和测试，展示软件在稳定性与安全性这两方面都符合要求。这一结论对于操作系统来说特别重要，因为这一结论表明，该操作系统的质量得到了认证，可以在任何应用中使用。

6.1.1任务
在µC/OS-II中，一个任务通常是一个无限的循环，就如大家编单任务程序的main函数。既然任务是一个无限循环，我们可以想象到它一定不会返回任何的数据，所以返回类型应该定义为void
  void YourTask(void *pdata)

  {

while(1)

{

    /*用户代码*/

    /*调用µC/OS的某些功能函数*/

    /*用户代码*/

}

}

注：pdata允许用户向该任务传递任何类型的参数；

任务函数永远不会返回。

每个任务有一个TCB（任务控制块），TCB里包含有任务的所有信息。µC/OS-II通过TCB感知任务的存在。

TCB的基本结构如下：

typedef struct os_tcb

{ OS_STK  *OSTCBStkPtr;        // 指向当前任务栈栈顶的指针

  struct os_tcb *OSTCBStkNext; // 指向任务的后一个任务

  struct os_tcb *OSTCBPrev;    // 指向任务的前一个任务

  INT16U OSTCBDly;             // 任务等待时钟节拍

  INT8U  OSTCBStat;            // 任务状态字

  INT8U  OSTCBPrio;            // 任务优先级

} OS_TCB;

6.1.2 任务状态

µC/OS-II中每个任务有5种状态，任意时刻每个任务处于且仅能处于某一种状态：睡眠态、就绪态、运行态、等待态、中断服务态。

睡眠态：指任务驻留在程序空间（ROM或RAM）中，还没有交给µC/OS-II管理，µC/OS-II还不知道任务的存在。通过调用OSTaskCreate()或OSTaskCreateExt()把任务交给µC/OS管理。

就绪态：任务经OSTaskCreate()或OSTaskCreateExt()建立后，就进入就绪态。可以通过调用OSTaskDel()把任务返回到睡眠态（任务本身或其它任务调用该函数）。

运行态：处于就绪态的任务当所有优先级比它高的任务都转为等待态时，就进入运行态。任何时刻有且仅有一个任务处于该状态。当所有用户任务都进入等待态时，系统任务——空闲任务OSTaskIdle进入运行态。

等待态：任务本身通过调用OSTimeDly(),OSTimeDlyHMSM(),OS**Pend()将自己延时一段时间或等待某个事件发生，进入该状态。

中断服务态：正在运行的任务是可以被中断的。响应中断时，正在执行的任务进入该状态，中断服务子程序控制CPU的使用权。

图6－1 任务状态转换图
6.1.3 任务调度

µC/OS-II是基于优先级抢占式调度的，任何时刻只要有高优先级任务进入就绪态，低优先级任务就将CPU让给高优先级任务。中断服务程序不是由µC/OS-II管理的，在中断服务程序里是不能进行任务调度的，相当于在响应中断时，调度器是关闭的，只有等中断返回时通过调用OSExitInt()函数才允许任务调度。

因为所有任务的TCB被链成一个双向链表，所以通过遍历这个双向链表很容易找到优先级最高的就绪任务，以实现任务调度。（可能我们按传统方法都这么做。）但是这样有一个问题，就是查找时间随任务的多少而变得不确定，当任务多时，查找时间长，任务少时，查找时间短。这不符合实时系统的时间确定性原则。

µC/OS-II是使用查表的方法找到优先级最高的就绪任务的（查表方法在µC/OS中被多处使用，这确实是一个不错的方法，使用又特别简单）。具体实现如下：
µC/OS-II中每一个任务的优先级是唯一的，不存在相同优先级的两个任务，所以可以根据优先级来唯一的标志任务。µC/OS-II支持64个任务，也就是有64个优先级0-63。µC/OS-II中将64个任务分成8组，用一个8位数据类型变量OSRdyGrp来表示8组中哪些组有就绪任务，若某一组中有任务处于就绪态，就将相应位置1，否则置0。每组8个任务，也用一个8位数据类型变量来表示8个任务中哪些任务处于就绪态，若有任务处于就绪态，就将相应位置1，否则置0。这8个8位数据类型变量被组织成一个数组OSRdyTbl[]。例如某个时刻，优先级为12的任务进入就绪态，这时就将OSRdyGrp的第1位置1，OSRdyGrp |= 0x02,将OSRdyTbl[1]的第4位置1，OSRdyTbl[1] |= 0x10。

这样假设任务的优先级为prio，prio的最高两位为0，因为最多只有64个任务，余下的6位的高3位就用来确定任务在哪个组，低3位就用来确定任务是组内哪个任务。

使任务进入就绪表的方法就可以这样：

OSRdyGrp |= OSMapTbl[prio>>3];
OSRdyTbl[prio>>3]  |= OSMapTbl[prio & 0x07];

OSMapTbl[]为一个预先建好的表，可以放在代码区，里面存有每个任务对应OSRdyGrp和OSRdyTbl[]的哪个位。

使任务脱离就绪表的方法为：
if( (OSRdyTbl[prio>>3] &= ~OSMapTbl[prio & 0x07]) == 0 )

    OSRdyGrp &= ~OSMapTbl[prio>>3];

找出优先级最高的就绪态任务的方法：
    y = OSUnMapTbl[OSRdyGrp];    // 找出优先级最高的组

    x = OSUnMapTbl[OSRdyTbl[y]];  // 找出优先级最高的组中优先级最高的任务

    prio = (y<<3) + x ; 

OSUnMapTbl[]是一个表，表长256，因为OSRdyGrp或OSRdyTbl[]值的范围为0~255。里面存有相应OSRdyGrp或OSRdyTbl[]对应优先级最高的任务。例如OSUnMapTbl[0x68] = 3，表示当OSRdyGrp=0x68时，最高任务在第3组，因为0x68=0b01101000，表示组3、5、6有任务处于就绪态，而组3的优先级最高。OSUnMapTbl[0x68] = 3，也表示当OSRdyTbl[y]= 0x68时，最高优先级的就绪任务的优先级为3，因为0x68=0b01101000，表示优先级为3、5、6的任务处于就绪态，而3的优先级最高。

用查表的方法实现任务的调度的一个突出的特点是：不管有多少任务，查找时间都固定。µC/OS就是用查表的方法，达到实时系统要求的。

任务的调度是通过调用OS_Sched()函数实现的。OS_Sched()的工作是找出处于就绪态且优先级最高的任务，然后做任务切换。任务切换是通过OS_TASK_SW()实现的。OS_TASK_SW()是跟特定处理器有关的函数，主要工作是保存原任务的状态，恢复新任务的状态，移植时须根据自己的处理器编写。实现任务切换还有另外一个函数OSIntCtxSw()，它是在中断服务函数返回时调用的。它的工作跟OS_TASK_SW()相似，只是可以比OS_TASK_SW()少掉保存原任务状态的代码，因为中断服务函数执行前硬件或编译器已经帮你保存了原任务的状态，但当前堆栈指针还是要保存在原任务TCB中的。

6.1.4 时间管理
µC/OS-II一共提供5个系统功能函数给用户做时间管理用：

OSTimeDly(),OSTimeDlyHMSM(),OSTimeDlyResume(),OSTimeGet(),OSTimeSet()。其中后面4个函数用户可以根据自己的需要决定是否编译它们，通过在OS_CFG.H中配置相应值，OS_TIME_DLY_HMSM_EN,OS_TIME_DLY_RESUME_EN,OS_TIME_GET_SET_EN,OS_TIME_GET_SET_EN

时间管理以一个函数OSTimeTick()为基础，OSTimeTick()的主要功能是：在每个时钟节拍将所有任务的等待时间OSTCBDly（在任务TCB中）减1，若有任务的OSTCBDly为0，且它又只是因为等待时间才没有进入就绪态，就使它进入就绪态。因此OSTimeTick()在内核的调用时机是在每次时钟中断服务函数里，移植时须设置一个时钟，在中断服务函数里调用OSTimeTick()。

OSTimeDly()的功能是将调用它的任务延时一段特定时间，这样调用它的任务将进入等待状态。

OSTimeDlyHMSM()的功能是将调用它的任务按时、分、秒、毫秒延时一段特定时间，这样调用它的任务将进入等待状态。要使用这个函数用户须将“每秒有几个时钟节拍”的信息告诉内核，通过配置OS_CFG.H文件中的OS_TICKS_PER_SEC的值。

OSTimeDlyResume()的功能是恢复被延时的任务，使它进入就绪态。

OSTimeGet()和OSTimeSet()的功能是得到系统时间和设置系统时间，系统时间是用一个32位变量OSTime表示的。OSTime在每个时钟节拍加1，当到最大值时回到0继续计数。

6.1.5 移植要求

要使µC/OS-II正常运行，处理器必须满足以下要求：

· 处理器的C编译器能产生可重入型代码；

· 处理器支持中断，并且能产生定时中断（通常为10～100Hz）；

· 用C语言就可以开/关中断；

· 处理器能支持一定数量的数据存储硬件堆栈（可能是几千字节）；

· 处理器有将堆栈指针以及其它CPU寄存器的内容读出、并存储到堆栈或内存中去的指令。

6.2 移植规划
6.2.1 编译器的选择

目前，针对ARM处理器核的C语言编译器有很多，如SDT、ADS、IAR、KEIL、TASKING和GCC等。据了解，目前在国内最流行的是ADS和GCC。ADS为ARM公司自己开发，GCC支持广泛，很多开发套件使用它作为编译器。

6.2.2 任务模式的取舍

ARM7TDMI处理器核具有用户、系统、管理、中止、未定义、中断和快中断七种模式。其中除用户模式外其它均为特权模式。管理、中止、未定义、中断和快中断与相应异常联系，不适合作为任何的工作模式，而系统模式除了是特权模式外，其他方面与用户模式一样，因而可选的任务模式只有用户模式和系统模式。为了尽量减少任务代码错误对整个程序的影响，缺省的任务模式定为用户模式，可选为系统模式。同时提供接口使任务可以在这两种模式间切换。

6.2.3 支持的指令集

ARM7TDMI处理器核具有两种指令集：标准32位ARM指令集和16位THUMB指令集。两种指令集有不同的应用范围。为了最大限度的支持芯片的特征，任务应当可以使用任意一种指令集并可以自由切换。而且不同的任务应当可以使用不同的指令集。

6.3 编写STR71x的启动代码


启动代码是芯片复位后进入C语言主函数前执行的一段代码，主要是为运行C语言程序提供基本运行环境，如初始化存储系统等。


ARM公司只设计处理器核，不自己生产芯片，只是把处理器核授权给其它芯片厂商，其它芯片厂商购买了授权后加入自己的外设，生产出各具特色的芯片。这样促进了基于ARM处理器核的芯片的多元化，但也使得各种芯片的启动代码差别很大，不易于编写出统一的启动代码（不像51核的芯片那样，51核的芯片的启动代码几乎一样，因此可以由编译器帮用户自动加入，这样用户可能无须关心启动代码）。ADS的策略是不提供完整的启动代码，启动代码不足部分或者由厂商提供，或者由用户自己编写。

本节介绍移植µC/OS-II时STR71x启动代码的编写。读者可参考71x_vect_ucos.s和71x_init.s两个源文件。71x_vect_ucos.s包括异常向量表及异常处理程序与C程序的接口；71x_init.s包括堆栈初始化、基本的系统初始化、变量初始化及跳转到C语言主函数。

6.3.1 异常向量表


异常是由内部或外部源（这里的内部与外部指的是从处理器核角度看的）产生的以引起处理器处理的一个事件。ARM处理器核支持7种类型的异常。异常出现后，CPU强制从异常类型对应的固定存储地址开始执行程序。这个固定的地址称为异常向量。


异常向量表的代码见程序清单6.1。这段代码位于文件71x_vect_ucos.s中。

程序清单6.1 异常向量表

vector_begin

        LDR     PC, Reset_Addr





（1）

        LDR     PC, Undefined_Addr




（2）

        LDR     PC, SWI_Addr





（3）

        LDR     PC, Prefetch_Addr




（4）

        LDR     PC, Abort_Addr





（5）

        NOP                             ; 保留

（6）

        LDR     PC, IRQ_Addr





（7）

        LDR     PC, FIQ_Addr





（8）

Reset_Addr     
DCD     Reset_Handler



（9）

Undefined_Addr  
DCD     UndefinedHandler


（10）

SWI_Addr       
DCD     SWIHandler




（11）

Prefetch_Addr    
DCD     PrefetchAbortHandler


（12）

Abort_Addr      
DCD     DataAbortHandler


（13）

               
DCD     0               ; 保留

（14）

IRQ_Addr      
DCD     OS_CPU_IRQ_ISR


（15）

FIQ_Addr      
DCD     OS_CPU_FIQ_ISR


（16）


向量从上到下依次为复位（程序清单（1））、未定义指令异常（程序清单（2））、软件中断（程序清单（3））、预取指中止（程序清单（4））、预取数据中止（程序清单（5））、保留的异常（程序清单（6））、IRQ（程序清单（7））和FIQ（程序清单（8））。使用LDR指令跳转而不使用B指令跳转的原因有两个：

（1） LDR指令可以全地址范围跳转而B指令不行；

（2） 芯片具有Remap功能，当向量表位于RAM中时，用B指令不能跳转到正确的位置。

6.3.2 异常处理程序与C程序的接口


虽然ADS可以使用__irq关键字声明一个C语言函数是用来处理中断的，从而避免使用汇编语言编写中断服务程序（由编译器自动生成完成保存寄存器及恢复寄存器的代码）。但是在移植µC/OS-II时不能这样处理，而必须使用汇编语言。这是因为使用C语言无法确保堆栈的结构，而所有的RTOS都必须使堆栈保持一定的结构。这部分代码见程序清单6.2。这段代码位于文件71x_vect_ucos.s中。

程序清单6.2 异常处理程序的汇编部分

        MACRO

        SaveContext $reg1,$reg2









（1）

        STMFD  sp!,{$reg1-$reg2,lr} 
;保存当前工作寄存器及

                              

;返回地址寄存器lr_ mode到堆栈中

        MRS    r1,spsr       

;将spsr_mode保存到r1中

        STMFD  sp!,{r1}       

;保存spsr寄存器到堆栈中

        MEND

        MACRO

        RestoreContext $reg1,$reg2









（2）

        LDMFD   sp!,{r1}        
;将spsr_mode寄存器恢复到r1中

        MSR     spsr_cxsf,r1    

;恢复spsr_mode寄存器

        LDMFD   sp!,{$reg1-$reg2,pc}^
;恢复当前工作寄存器
                                
;返回到异常中断的下一条指令

        MEND

UndefinedHandler

        SaveContext r0,r11    


;保存当前工作寄存器及
                              

;lr_ und和spsr_und到堆栈中

        BL      Undefined_Handler
;跳转到C语言函数Undefined_Handler
        RestoreContext r0,r11 


;返回到未定义指令的下一条指令

SWIHandler

        SaveContext r0,r11    


;保存当前工作寄存器及
                              

;lr_ svc和spsr_svc到堆栈中

        BL       SWI_Handler  

;跳转到C语言函数SWI_Handler.

        RestoreContext r0,r11 


;返回到SWI指令的下一条指令

PrefetchAbortHandler

        SUB    lr,lr,#4       

;更新连接寄存器lr

        SaveContext r0,r11    


;保存当前工作寄存器及
                              

;lr_abt和spsr_abt到堆栈中

        BL     Prefetch_Handler

;跳转到C语言函数Prefetch_Handler.

        RestoreContext r0,r11 


;返回到产生预取指中止异常的指令的下一条指令

DataAbortHandler

        SUB    lr,lr,#8       

;更新连接寄存器lr
        SaveContext r0,r11    


;保存当前工作寄存器及
                              

;lr_abt和spsr_abt到堆栈中
        BL     Abort_Handler  

;跳转到C语言函数Abort_Handler.

        RestoreContext r0,r11 


;返回到产生数据中止异常的指令的下一条指令
       MACRO

       IRQ_to_SYS

        MSR    cpsr_c,#0x1F   

;切换到系统模式
        STMFD  sp!,{lr}       

;保存连接寄存器lr到堆栈中

       MEND

       MACRO

        SYS_to_IRQ

        LDMFD  sp!,{lr}      

;恢复连接寄存器lr

        MSR    cpsr_c,#0xD2  

;切换到IRQ模式

        MOV    pc,lr         

;返回

       MEND

T0TIMIIRQHandler

        IRQ_to_SYS

        BL     T0TIMI_IRQHandler

        SYS_to_IRQ

FLASHIRQHandler

        IRQ_to_SYS

        BL     FLASH_IRQHandler

        SYS_to_IRQ

RCCUIRQHandler

        IRQ_to_SYS

        BL     RCCU_IRQHandler

        SYS_to_IRQ

RTCIRQHandler

        IRQ_to_SYS

        BL     RTC_IRQHandler

        SYS_to_IRQ

WDGIRQHandler

        IRQ_to_SYS

        BL     WDG_IRQHandler

        SYS_to_IRQ

XTIIRQHandler

        IRQ_to_SYS

        BL     XTI_IRQHandler

        SYS_to_IRQ

USBHPIRQHandler

        IRQ_to_SYS

        BL     USBHP_IRQHandler

        SYS_to_IRQ

I2C0ITERRIRQHandler

        IRQ_to_SYS

        BL     I2C0ITERR_IRQHandler

        SYS_to_IRQ

I2C1ITERRIRQHandler

        IRQ_to_SYS

        BL     I2C1ITERR_IRQHandler

        SYS_to_IRQ

UART0IRQHandler

        IRQ_to_SYS

        BL     UART0_IRQHandler

        SYS_to_IRQ

UART1IRQHandler

        IRQ_to_SYS

        BL     UART1_IRQHandler

        SYS_to_IRQ

UART2IRQHandler

        IRQ_to_SYS

        BL     UART2_IRQHandler

        SYS_to_IRQ

UART3IRQHandler

        IRQ_to_SYS

        BL     UART3_IRQHandler

        SYS_to_IRQ

BSPI0IRQHandler

        IRQ_to_SYS

        BL     BSPI0_IRQHandler

        SYS_to_IRQ

BSPI1IRQHandler

        IRQ_to_SYS

        BL     BSPI1_IRQHandler

        SYS_to_IRQ

I2C0IRQHandler

        IRQ_to_SYS

        BL     I2C0_IRQHandler

        SYS_to_IRQ

I2C1IRQHandler

        IRQ_to_SYS

        BL     I2C1_IRQHandler

        SYS_to_IRQ

CANIRQHandler

        IRQ_to_SYS

        BL     CAN_IRQHandler

        SYS_to_IRQ

ADC12IRQHandler

        IRQ_to_SYS

        BL     ADC12_IRQHandler

        SYS_to_IRQ

T1TIMIIRQHandler

        IRQ_to_SYS

        BL     T1TIMI_IRQHandler

        SYS_to_IRQ

T2TIMIIRQHandler

        IRQ_to_SYS

        BL     T2TIMI_IRQHandler

        SYS_to_IRQ

T3TIMIIRQHandler

        IRQ_to_SYS

        BL     T3TIMI_IRQHandler

        SYS_to_IRQ

HDLCIRQHandler

        IRQ_to_SYS

        BL     HDLC_IRQHandler

        SYS_to_IRQ

USBLPIRQHandler

        IRQ_to_SYS

        BL     USBLP_IRQHandler

        SYS_to_IRQ

T0TOIIRQHandler

        IRQ_to_SYS

        BL     T0TOI_IRQHandler

        SYS_to_IRQ

T0OC1IRQHandler

        IRQ_to_SYS

        BL     T0OC1_IRQHandler

        SYS_to_IRQ

T0OC2IRQHandler

        IRQ_to_SYS

        BL     T0OC2_IRQHandler

        SYS_to_IRQ
程序定义宏SaveContext（程序清单6.2（1））和宏RestoreContext（程序清单6.2（2）），分别用于保存现场和恢复现场。宏SaveContext将从reg1到reg2的寄存器及lr_ mode和spsr_mode寄存器的值保存到堆栈中。宏RestoreContext从堆栈中恢复从reg1到reg2的寄存器及lr_ mode和spsr_mode寄存器的值。

对每个异常处理程序都首先调用宏SaveContext保存异常现场，然后跳转到C语言的异常处理程序，最后调用宏RestoreContext恢复异常现场。其中对于预取指中止异常和数据中止异常，在调用宏SaveContext之前首先要更新连接寄存器，这是因为这两个异常的返回地址存在一个偏移值。

对于IRQ异常，这里使用向量形式的异常处理程序，使每个中断源对应一个处理程序。首先调用宏IRQ_to_SYS（程序清单6.2（3））将处理器从IRQ模式切换到系统模式，并将连接寄存器保存到堆栈中，然后跳转到C语言的异常处理程序，最后调用宏SYS_to_IRQ（程序清单6.2（4））恢复连接寄存器，并将处理器从系统模式切换到IRQ模式及中断返回。

6.3.3 堆栈初始化


因为ARM有7种执行模式，每一种模式的堆栈指针寄存器（SP）都是独立的（除了用户模式和系统模式使用同一个堆栈指针寄存器外）。因此，对程序中需要用到的每种模式都要给相应的SP定义一个堆栈地址。实现这个功能的方法是改变程序状态寄存器CPSR的模式位，使处理器切换到不同的模式，然后给相应的SP赋值。这部分功能的实现代码见程序清单6.1。这段代码位于文件71x_init.s中。

程序清单6.1 堆栈初始化

RAM_Base          
EQU   
0x20000000

RAM_Limit        
EQU   
0x20010000
Stack_Base          
EQU   
RAM_Limit

SVC_Stack_Length   
EQU   
256

IRQ_Stack_Length   
EQU   
1024

USR_Stack_Length   
EQU   
1024

FIQ_Stack_Length   
EQU   
256

ABT_Stack_Length   
EQU   
256

UNDEF_Stack_Length 
EQU   
256

SVC_Stack          
EQU    Stack_Base                 
; SVC stack

IRQ_Stack          
EQU    SVC_Stack-SVC_Stack_Length 
; followed by IRQ stack

USR_Stack          
EQU    IRQ_Stack-IRQ_Stack_Length 
; followed by USR stack

FIQ_Stack          
EQU    USR_Stack-USR_Stack_Length 
; followed by FIQ stack

ABT_Stack         
EQU    FIQ_Stack-FIQ_Stack_Length 
; followed by ABT stack

UNDEF_Stack       
EQU    ABT_Stack-ABT_Stack_Length 
; followed by UNDEF stack
        MSR     CPSR_c, #Mode_ABT:OR:F_Bit:OR:I_Bit

（1）

        LDR     SP, =ABT_Stack







（2）

        MSR     CPSR_c, #Mode_UNDEF:OR:F_Bit:OR:I_Bit

（3）

        LDR     SP,= UNDEF_Stack






（4）

        MSR     CPSR_c, #Mode_SVC:OR:F_Bit:OR:I_Bit

（5）

        LDR     SP, =SVC_Stack







（6）

       
MSR     CPSR_c, #Mode_FIQ:OR:F_Bit:OR:I_Bit


（7）

        LDR     SP, =FIQ_Stack







（8）

       
MSR     CPSR_c, #Mode_IRQ:OR:F_Bit:OR:I_Bit


（9）

        LDR     SP, =IRQ_Stack







（10）

        MSR     CPSR_c, #Mode_SYS:OR:F_Bit:OR:I_Bit


（11）

        LDR     SP, =USR_Stack







（12）


程序首先把处理器切换到中止模式（程序清单6.1（1）），并设置中止模式的堆栈指针（程序清单6.1（2））。其中ABT_Stack为分配给中止模式的堆栈空间的起始地址，ABT_Stack_Length为分配给中止模式的堆栈空间的大小。


程序使用同样的方法设置未定义模式堆栈指针（程序清单6.1（3）、（4）），管理模式堆栈指针（程序清单6.1（5）、（6）），FIQ模式堆栈指针（程序清单6.1（7）、（8）），IRQ模式堆栈指针（程序清单6.1（9）、（10）），用户模式堆栈指针（程序清单6.1（11）、（12））。由于系统模式与用户模式使用同一个堆栈指针寄存器，因此就不需要再设置系统模式的堆栈指针了。


这段程序执行完以后，处理器处于用户模式下。在跳转到C语言主函数前将处理器从用户模式切换到系统模式。

6.3.4 基本的系统初始化


为了使系统基本能够工作，必须在进入C语言主函数前对系统进行一些基本的初始化工作，这些工作用户可根据需要进行修改。代码见程序清单6.2。这段代码位于文件71x_init.s中。

程序清单6.2 系统基本初始化

        MACRO   

        PERIPHERAL_INIT










（1）

        LDR     r1, =APB1_base_addr      
; r0= APB1基地址

        LDR     r2, =APB2_base_addr      
; r0= APB2基地址
        LDR     r0, =CKDIS1_config_all

        STRH    r0, [r1, #CKDIS_off_addr]

;禁用所有APB1外设的时钟

        LDR     r0, =CKDIS2_config_all

        STRH    r0, [r2, #CKDIS_off_addr]

;禁用所有APB2外设的时钟
        LDR     r0, =SWRES1_config_all

        STRH    r0, [r1, #SWRES_off_addr]

;复位所有APB1外设
        LDR     r0, =SWRES2_config_all

        STRH    r0, [r2, #SWRES_off_addr]

;复位所有APB2外设
        MOV     r7, #10                  
;等待复位退出

loop1   
SUBS    r7, r7, #1

        BNE     loop1

        MOV     r0, #0

        STRH    r0, [r1, #SWRES_off_addr]

;使能所有APB1外设
        STRH    r0, [r2, #SWRES_off_addr] 
;使能所有APB2外设
        STRH    r0, [r1, #CKDIS_off_addr]

;使能所有APB1外设的时钟
        STRH    r0, [r2, #CKDIS_off_addr]

;使能所有APB2外设的时钟
        MOV     r7, #10                  
;等待复位退出
loop2   
SUBS    r7, r7, #1

        BNE     loop2

        MEND
        MACRO   

        EMI_INIT












（2）

        LDR     r0, =GPIO2_Base_addr      
;将P2.0- P2.3 配置成AF_PP模式

        LDR     r2, [r0, #PC0_off_addr]

        ORR     r2, r2,#0x0000000F

        STR     r2, [r0, #PC0_off_addr]

        LDR     r2, [r0, #PC1_off_addr]

        ORR     r2, r2,#0x0000000F

        STR     r2, [r0, #PC1_off_addr]

        LDR     r2, [r0, #PC2_off_addr]

        ORR     r2, r2,#0x0000000F

        STR     r2, [r0, #PC2_off_addr]

        LDR     r0, =EMI_Base_addr

        LDR     r1, =0x18:OR:EMI_ENABLE:OR:EMI_SIZE_16

        STR     r1, [r0, #BCON1_off_addr] 

;使能Bank1，16-bit，7个等待状态

        MEND
        MACRO   

        EIC_INIT












（3）

        LDR     r3, =EIC_Base_addr

        MOV     r0, #0

        MVN     r1, #0

        MOV     r2, #0x0c

        STR     r0, [r3, #ICR_off_addr]

        STR     r0, [r3, #IER_off_addr]

        STR     r1, [r3, #IPR_off_addr]

        STR     r2, [r3, #FIR_off_addr]

        STR     r0, [r3, #CIPR_off_addr]

        LDR     r4, =0xE59F0000

        STR     r4, [r3, #IVR_off_addr]

;将指令LDR pc,[pc,#offset]写到IVR[31:16]中

        LDR     r2, =32                
;有32个寄存器待初始化
        LDR     r0, =T0TIMI_Addr       
;读取T0TIMI_Addr向量地址

        LDR     r1, =0x00000FFF

        AND     r0,r0,r1

        LDR     r5, =SIR0_off_addr  
;读取SIR0地址

        SUB     r4,r0,#8           
;因为预取指，减去8

        LDR     r1, =0xF7E8       
;加上偏移量

                                  
;0xF7E8用于构成LDR pc,[pc,#offset]的操作码

        ADD     r1,r4,r1           
;计算跳转偏移

EIC_INI 
MOV     r4, r1, LSL #16     
;结果左移

        STR     r4, [r3, r5]           ;将结果保存到SIRx寄存器中

        ADD     r1, r1, #4          
;下一个IRQ地址

        ADD     r5, r5, #4          
;下一个SIR寄存器

        SUBS    r2, r2, #1           
;递减待初始化的SIR寄存器数目

        BNE     EIC_INI           
;如果数目不为零，继续

        MEND
       
PERIPHERAL_INIT         
;复位所有外设

        EMI_INIT                  
;初始化EIM Bank 1
        EIC_INIT                  
;初始化EIC
  IF :DEF: remapping

IF :DEF: remap_flash

        MOV     r0, #FLASH_mask

   
ENDIF

   
IF :DEF: remap_ram

        MOV     r0, #RAM_mask

   
ENDIF

   
IF :DEF: remap_ext

        MOV     r0, #EXTMEM_mask

    ENDIF

        LDR     r1, =CPM_Base_addr

        LDRH    r2, [r1, #BOOTCR_off_addr]
;读取BOOTCR寄存器

        BIC     r2, r2, #0x03             

;清除BOOTCR的最后两位

        ORR     r2, r2, r0                
;修改BOOTCR的最后两位
        STRH    r2, [r1, #BOOTCR_off_addr]
;写BOOTCR寄存器

  ENDIF
程序定义了三个宏，宏PERIPHERAL_INIT（程序清单6.2（1））用于复位所有的外设；宏EMI_INIT（程序清单6.2（2））用于初始化EMI Bank 1；宏EIC_INIT（程序清单6.2（3））将EIC初始化为：禁用IRQ和FIQ，IVR包含装载PC指令操作码（0xF59FF00），当前优先级别为0，禁用所有通道，所有通道的优先级为0，所有的SIR寄存器包含相应IRQ向量地址。程序清单6.2（4）修改BOOTCR寄存器以实现存储器重映射。

6.3.5 变量初始化


映像一开始总是存储在ROM/Flash里面的（虽然调试的时候可以将映像放在RAM中，但这部分RAM也是看作ROM用的），其RO输出段既可以在ROM/Flash里面执行，也可以转移到速度更快的RAM中执行；而RW和ZI这两个输出段是必须转移到可写的RAM里去的。所谓变量初始化就是指完成必要的从ROM到RAM的数据传输（对RW输出段）和内容清零（对ZI输出段）。关于RO、RW和ZI输出段读者可参考ADS连接器的相关文档。


下面是在ADS环境下，连接器导出的一些常用符号：


Image$$RO$$Base

RO输出段的起始地址，即连接选项里ro_base指定的地址；


Image$$RO$$Limit
RO输出段结束地址的下一个地址；

Image$$RW$$Base

RW输出段的起始地址，即连接选项里rw_base指定的地址；


Image$$RW$$Limit
ZI输出段结束地址的下一个地址（为了保持跟以前代码的兼容，ADS将这个符号定为ZI输出段结束地址的下一个地址，而不是RW输出段结束地址的下一个地址）；


Image$$ZI$$Base

ZI输出段的起始地址；


Image$$ZI$$Limit

ZI输出段结束地址的下一个地址。

虽然在ADS环境下，用户可使用跳转到__main函数，然后再由__main跳转到C语言main的方式让编译器自动生成完成变量初始化功能的代码。这里我们选择直接实现的方式，代码见程序清单6.3。这段代码位于文件71x_init.s中。

程序清单6.3 变量初始化


    IMPORT
|Image$$RO$$Limit|  
;RW输出段的初始数据在ROM里的起始地址

    IMPORT
|Image$$RW$$Base|  
;RW输出段在RAM里的起始地址

    IMPORT
|Image$$ZI$$Base|  
;ZI输出段在RAM里的起始地址

    IMPORT
|Image$$ZI$$Limit| 
;ZI输出段结束地址的下一个地址

    LDR
    r0, =|Image$$RO$$Limit|

    LDR
    r1, =|Image$$RW$$Base|


    LDR
    r3, =|Image$$ZI$$Base|

    CMP
    r0, r1
    
;比较它们是否相等


    BEQ
    %F1










（1）

0















（2）


    CMP
    r1, r3
    
;复制RW初始数据


    LDRCC   r2, [r0], #4    

;即LDRCC r2, [r0]  
ADD r0, r0, #4

 


    STRCC   r2, [r1], #4    

;即STRCC r2, [r1] 
ADD r1, r1, #4


    BCC
    %B0









（3）

1















（4）


    LDR
    r1, =|Image$$ZI$$Limit|


    MOV
    r2, #0

2















（5）


    CMP
    r3, r1
    
;清零操作

    STRCC   r2, [r3], #4


    BCC
    %B2









（6）


程序先把ROM里Image$$RO$$Limit开始的RW初始数据复制到RAM里Image$$RW$$Base开始的地址，当RAM这边的目标地址到达Image$$ZI$$Base后就表示RW输出段的结束和ZI输出段的开始，接下去就对ZI输出段进行清零操作，直至遇到ZI输出段结束地址的下一个地址Image$$ZI$$Limit。上述程序段用到了局部标号（程序清单6.3（2）、（4）、（5））即其引用格式（程序清单6.3（1）、（3）、（6）），读者可参见ARM伪指令的相关文档。

6.3.6 跳转到C语言主函数


启动代码实现的最后功能是跳转到C语言主函数，这个功能的最简单实现是：

        IMPORT  main

        B       main
;main为C语言主函数入口


直接从启动代码跳转到C语言主函数入口，当然C语言主函数入口可以由用户随便定义。


如6.3.5节中所介绍的，ADS环境还提供了一套系统级的转移机制：

        IMPORT  __main

        B       __main
__main是编译器提供的一个函数，负责完成库函数的初始化和变量初始化。__main最后自动跳转到main()函数。


这部分功能的实现代码见程序清单6.4。这段代码位于文件71x_init.s中。

程序清单6.4 跳转到C语言主函数

        MSR     CPSR_c, #Mode_SYS



（1）

        IMPORT  Main

        B       Main

程序先把处理器从用户模式切换到系统模式（程序清单6.4（1）），然后直接跳转到C语言主函数Main。

6.4 移植µC/OS-II


µC/OS-II的代码被分成3个部分：与处理无关的代码；与应用相关的代码；与处理器相关的代码。移植时只要考虑与处理器相关的代码。与处理器相关的代码由3个文件组成：OS_CPU.H文件；OS_CPU_C.C文件；OS_CPU_A.S文件。本节描述顺序基本按照《嵌入式实时操作系统µC/OS-II（第2版）》第13章介绍的移植步骤。
6.4.1 OS_CPU.H文件


OS_CPU.H文件中包含与处理器和编译器相关的常量、宏的定义。

6.4.1.1 数据类型

µC/OS-II不直接使用C语言的short、int、long等数据类型的定义，因为它们与处理器类型和编译器相关，不利于移植；而使用typedef或预定义的方式重新定义了数据类型。这样既有利于移植，又很直观。为了完成数据类型定义的移植，应该参考所使用的C编译器的文档，从中找到其使用的数据类型的定义，从而使µC/OS-II得到正确的数据类型定义。根据ADS编译器的特征，这些代码如程序清单6.5所示。

程序清单6.5 数据类型

typedef unsigned char  
BOOLEAN; 



（1）

typedef unsigned char  
INT8U; 

typedef signed char  
INT8S; 

typedef unsigned short 
INT16U; 

typedef signed short 
INT16S; 

typedef unsigned long  
INT32U; 

typedef signed long  
INT32S; 

typedef float         
FP32; 




（2）

typedef double       
FP64; 

typedef unsigned int   
OS_STK; 



（3）

typedef unsigned int   
OS_CPU_SR; 


（4）


程序清单6.5（1）定义了可移植且直观的整型数据类型。举例来说，INT16U数据类型总是代表16位的无符号整型数。这样，µC/OS-II和应用程序就可以断定，声明为该数据类型变量的范围都是0～65535。

程序清单6.5（2）是浮点数据类型的定义，虽然µC/OS-II中并不使用浮点数，但是为方便起见，这里还是定义了浮点数据类型。

必须将任务堆栈的数据类型告诉µC/OS-II。这是通过为OS_STK声明恰当的C数据类型来实现的。ARM处理器的堆栈为32位，程序清单6.5（3）将OS_STK声明为32位整型。

OS_CPU_SR为CPU状态寄存器的数据类型，ARM处理器的CPSR和SPSR均为32位，程序清单6.5（4）将OS_CPU_SR声明为32位整型。

6.4.1.2 OS_ENTER_CRITICAL()和OS_EXIT_CRITICAL()
µC/OS-II使用宏OS_ENTER_CRITICAL()和OS_EXIT_CRITICAL()实现临界段代码的保护。这两个宏与处理器和编译器有关，需要根据自己的处理器和编译器进行移植。代码见程序清单6.6。

程序清单6.6 宏OS_ENTER_CRITICAL()和OS_EXIT_CRITICAL()

OS_CPU_SR_Save














（1）

        MRS     R0,CPSR         
;设置CPSR 的IRQ和FIQ位以关中断
        ORR     R1,R0,#NO_INT

        MSR     CPSR_c,R1

        MRS     R1,CPSR       

;确认CPSR包含正确的中断屏蔽位

        AND     R1,R1,#NO_INT

        CMP     R1,#NO_INT

        BNE     OS_CPU_SR_Save  
;没有关掉中断，再试一下

        BX      LR               
;成功关了中断，返回原来CPSR值

OS_CPU_SR_Restore













（2）

        MSR     CPSR_c,R0


;用传入的值恢复CPSR_c

        BX      LR
#define  OS_CRITICAL_METHOD    3









（3）

#define  OS_ENTER_CRITICAL()  
{cpu_sr = OS_CPU_SR_Save();}



（4）

#define  OS_EXIT_CRITICAL()   
{OS_CPU_SR_Restore(cpu_sr);}



（5）

µC/OS-II支持3种不同的方式实现宏OS_ENTER_CRITICAL()和OS_EXIT_CRITICAL()，只需要其中一种方式就可以了。这里使用第3种方式，由OS_CRITICAL_METHOD常量指出（程序清单6.6（3））。OS_ENTER_CRITICAL()调用

OS_CPU_SR_Save()函数（程序清单6.6（1），实现将CPSR的IRQ和FIQ中断屏蔽位置1，并将OS_CPU_SR_Save()函数的返回值保存到全局变量cpu_sr中（程序清单6.6（4））。OS_EXIT_CRITICAL()将全局变量cpu_sr保存的值作为参数调用OS_CPU_SR_Restore(cpu_sr)函数（程序清单6.6（2）），实现将CPSR恢复成cpu_sr保存的值。程序清单6.6（1）和程序清单6.6（2）位于文件OS_CPU_A.S中。

6.4.1.3 OS_STK_GROWTH

µC/OS-II需要知道堆栈的增长方向，这是因为：首先，OSInit()需要知道，当用OS_TaskIdle()和OS_TaskStat()函数建立任务时，堆栈的顶端地址在哪里；其次，在调用OSTaskStkChk()函数时，µC/OS-II需要知道堆栈的底端地址在哪里，从而得到堆栈的使用情况。
绝大多数处理器的堆栈是从上往下递减的，但是也有某些处理器使用的是相反的方向。µC/OS-II被设计成对两种情况都可以支持，移植时只要配置OS_STK_GROWTH常数以指定堆栈的增长方向就可以了。置OS_STK_GROWTH为0，表示堆栈从下（低地址）往上（高地址）递增；置OS_STK_GROWTH为1，表示堆栈从上（高地址）往下（低地址）递减。

虽然ARM处理器对于两种方式都支持，但ADS的C语言编译器仅支持一种方式，即从上往下增长，并且必须是满递减堆栈，所以移植时将OS_STK_GROWTH定义为1。代码见程序清单6.7。

程序清单6.7 OS_STK_GROWTH定义

#define  OS_STK_GROWTH        1

程序实现将OS_STK_GROWTH定义为1。

6.4.1.4 OS_TASK_SW()

OS_TASK_SW()函数是µC/OS-II从低优先级任务切换到最高优先级任务时调用的。OS_TASK_SW()总是在任务级代码中被调用；另一个函数OSIntExit()被用来在ISR使得更高优先级任务处于就绪状态时，执行任务切换功能。任务切换将处理器的寄存器保存到将被挂起的任务的堆栈中，并且从堆栈中恢复即将要运行的更高优先级的任务。


在µC/OS-II中，处于就绪态任务的堆栈结构看起来就像刚刚发生过中断一样，所有的寄存器都保存在堆栈中。换句话说，µC/OS-II要运行处于就绪态的任务必须要做的事就是，从任务堆栈中恢复处理器所有的寄存器，并且执行中断返回指令。为了实现任务切换，可以通过执行OS_TASK_SW()模仿中断的产生。绝大多数处理器会提供软中断或指令陷阱来完成这个功能；该中断服务程序或指令陷阱处理函数的中断向量地址必须指向汇编函数OSCtxSw()。


这里我们用宏定义的方法将OS_TASK_SW()简单定义为OSCtxSw()，代码见程序清单6.8。

程序清单6.8 OS_TASK_SW()宏

#define  OS_TASK_SW()

OSCtxSw()
程序实现将OS_TASK_SW()定义为OSCtxSw()。

6.4.2 OS_CPU_C.C文件

µC/OS-II的移植范例要求用户编写10个简单的C函数：

OSTaskStkInit();

OSTaskCreateHook();

OSTaskDelHook();

OSTaskSwHook();

OSTaskIdleHook();

OSTaskStatHook();

OSTimeTickHook();

OSInitHookBegin();

OSInitHookEnd();

OSTCBInitHook();


唯一必要的函数是OSTaskStkInit()，其它9个函数必须声明，但并不一定要包含任何代码。用户可根据自己的需要进行修改。如果用户定义了…Hook函数，则必须将OS_CPU_HOOKS_EN定义为1。代码见程序清单6.9。

程序清单6.9 OS_CPU_HOOKS_EN定义

#define OS_CPU_HOOKS_EN           1
程序实现将OS_CPU_HOOKS_EN定义为1。

6.4.2.1 任务堆栈初始化函数OSTaskStkInit()

函数OSTaskCreate()和OSTaskCreateExt()通过调用OSTaskStkInit()，初始化任务的栈结构；因此堆栈看起来就像中断刚发生过一样，所有寄存器都保存在堆栈中。OSTaskStkInit()的示意性代码如下所示：

OS_STK *OSTaskStkInit (void (*task)(void *pd), void *pdata, OS_STK *ptos, INT16U opt)

{


模拟带参数（pdata）的函数调用；


模拟ISR向量；


按照预先设计的寄存器值初始化堆栈结构；


返回栈顶指针给调用该函数的函数；

}

移植时使用的堆栈结构如图6－2所示，代码见程序清单6.10。























图6－2 任务堆栈结构

程序清单6.10 函数OSTaskStkInit()

OS_STK *OSTaskStkInit (void (*task)(void *pd), void *p_arg, OS_STK *ptos, INT16U opt)

{

    OS_STK *stk;

    opt 

= opt;                 

/* opt没有使用，避免编译警告 */

    stk    
= ptos;          



/* 获取堆栈指针 */

    *(stk) 
= (OS_STK)task;



/* PC，任务入口点 */



（1）

    *(--stk) 
= (INT32U)0x13131313L; 

/* R13(LR) 
*/




（2）

    *(--stk) 
= (INT32U)0x12121212L;     
/* R12  

*/

    *(--stk) 
= (INT32U)0x11111111L;     
/* R11    
*/

    *(--stk) 
= (INT32U)0x10101010L;     
/* R10   
*/

    *(--stk) 
= (INT32U)0x09090909L;     
/* R9    
*/

    *(--stk) 
= (INT32U)0x08080808L;     
/* R8  

*/

    *(--stk) 
= (INT32U)0x07070707L;     
/* R7  

*/

    *(--stk)
= (INT32U)0x06060606L;     
/* R6   

*/

    *(--stk) 
= (INT32U)0x05050505L;     
/* R5    
*/

    *(--stk) 
= (INT32U)0x04040404L;     
/* R4   

*/

    *(--stk) 
= (INT32U)0x03030303L;     
/* R3 

*/

    *(--stk) 
= (INT32U)0x02020202L;     
/* R2  

*/

    *(--stk) 
= (INT32U)0x01010101L;     
/* R1  

*/

    *(--stk) 
= (INT32U)p_arg;           
/* R0，第一个参数使用R0传递 */
（3）

    *(--stk) 
= (INT32U)ARM_SVC_MODE;
/* CPSR，（允许IRQ和FIQ中断） */

    return (stk);














（4）

}

OSTaskStkInit()模仿调用应用程序任务代码的情况，将任务起始地址保存到堆栈中（程序清单6.10（1））；同样，OSTaskStkInit()将寄存器R13-R1保存到堆栈中，这里将它们分别设定一个值（程序清单6.10（2））；因为ADS编译器是使用寄存器R0传递第一个参数的，所以将p_arg保存到堆栈中R0的位置（程序清单6.10（3））；在初始化堆栈后，OSTaskStkInit()应当返回堆栈指针指向的地址（程序清单6.10（4）），OSTaskCreate()或OSTaskCreateExt()得到这个地址并且保存在任务控制块中，再次强调，因为ADS编译器使用的是满堆栈，所以返回的地址指向的是最后一个压栈的值，而不是下一个空的堆栈空间。

6.4.2.2 任务建立接口函数OSTaskCreateHook()

每当添加任务时，OS_TCBInit()函数都会调用OSTaskCreateHook()函数，该函数允许用户扩展µC/OS-II的功能。在OS_TCBInit()函数中，当µC/OS-II完成任务控制块（OS_TCB）初始化的绝大部分工作后，但在任务控制块被链接到相应的任务链中之前，以及在该任务就绪运行之前，µC/OS-II会调用OSTaskCreateHook()函数。该函数被调用时中断是开着的。


当OSTaskCreateHook()被调用时，它会收到指向刚刚建立任务的任务控制块的指针。这样，OSTaskCreateHook()就可访问该任务控制块结构所有的成员了。当使用OSTaskCreate()建立任务时，OSTaskCreateHook()的功能是有限的；但当使用OSTaskCreateExt()建立任务时，OSTaskCreateHook()会得到OS_TCB中的扩展指针（OSTCBExtPtr），该指针可用来访问任务的附加数据，如任务计数器以及调试信息等。


这里我们将OSTaskCreateHook()写成空函数，代码见程序清单6.11。

程序清单6.11 函数OSTaskCreateHook()

#if OS_CPU_HOOKS_EN > 0

void  OSTaskCreateHook (OS_TCB *ptcb)

{
    ptcb = ptcb;          
/* 防止编译警告 */
}

#endif

程序将OSTaskCreateHook()定义为空函数。

6.4.2.3 任务删除接口函数OSTaskDelHook()

在任务从就绪列表或等待列表（若任务在等待某事件发生）中被删除后，OSTaskDel()函数就会调用OSTaskDelHook()函数。该函数在将任务从µC/OS-II的内部有效任务链表中删除之前被调用。当调用OSTaskDelHook()时，OSTaskDelHook()会收到一个指向正在被删除任务的任务控制块的指针。这样它就可以访问该任务控制块所有的结构成员了。OSTaskDelHook()可以用来检验TCB扩展部分是否已建立了（一个非空的指针），并进行一些清零操作等。OSTaskDelHook()被调用时，中断是关掉的，所以该函数执行太长会影响中断响应时间。


这里我们将OSTaskDelHook()写成空函数，代码见程序清单6.12。

程序清单6.12 函数OSTaskDelHook()

#if OS_CPU_HOOKS_EN > 0

void  OSTaskDelHook (OS_TCB *ptcb)

{

    ptcb = ptcb;       
/* 防止编译警告 */
}

#endif
程序将OSTaskDelHook()定义为空函数。

6.4.2.4 任务切换接口函数OSTaskSwHook()

µC/OS-II在做任务切换时，会调用OSTaskSwHook()函数。不管任务切换是通过OSCtxSw()函数（见6.4.3节：OS_CPU_A.S文件）实现的，还是通过OSIntCtxSw()函数实现的，都会调用该函数。OSTaskSwHook()可以直接访问OSTCBCur和OSTCBHighRdy这两个全局变量，OSTCBCur指向将被切换出去的任务的任务控制块，而OSTCBHighRdy指向新任务的任务控制块。OSTaskSwHook()没有任何参数，也不返回任何值。OSTaskSwHook()被调用时，中断是关掉的，所以该函数执行太长会影响中断响应时间。


这里我们将OSTaskSwHook ()写成空函数，代码见程序清单6.13。

程序清单6.13 函数OSTaskSwHook ()

#if (OS_CPU_HOOKS_EN > 0) && (OS_TASK_SW_HOOK_EN > 0)

void  OSTaskSwHook (void)

{
}

#endif
程序将OSTaskSwHook ()定义为空函数。

6.4.2.5 空闲任务接口函数OSTaskIdleHook()

OSTaskIdleHook()函数在OSTaskIdle()函数中被调用。很多微处理器都允许执行相应的指令，将CPU置于低功耗模式，而当接收到中断信号时，CPU就会退出低功耗模式。OSTaskIdleHook()函数可用于实现CPU的这种低功耗模式。OSTaskIdleHook()没有任何参数，也不返回任何值。


这里我们将OSTaskIdleHook ()写成空函数，代码见程序清单6.14。

程序清单6.14 函数OSTaskIdleHook ()

#if OS_CPU_HOOKS_EN > 0 && OS_VERSION >= 251

void  OSTaskIdleHook (void)

{

}

#endif
程序将OSTaskIdleHook ()定义为空函数。

6.4.2.6 统计任务接口函数OSTaskStatHook()

OSTaskStatHook()函数每秒钟都会被统计任务OSTaskStat()调用一次。用户可以用OSTaskStatHook()扩展统计任务的功能。例如，可以跟踪并显示每个任务的执行时间、每个任务所使用的CPU时间以及每个任务执行的频率等。OSTaskStatHook()没有任何参数，也不返回任何值。


这里我们将OSTaskStatHook()写成空函数，代码见程序清单6.15。

程序清单6.15 函数OSTaskStatHook()

#if OS_CPU_HOOKS_EN > 0

void  OSTaskStatHook (void)

{

}

#endif
程序将OSTaskStatHook()定义为空函数。

6.4.2.7 时钟节拍接口函数OSTimeTickHook()

OSTimeTickHook()函数在每个时钟节拍都会被OSTimeTick()调用一次。实际上，OSTimeTickHook()是在µC/OS-II真正处理时钟节拍之前被调用的，以便于用户能先处理应急的事务。OSTimeTickHook()没有任何参数，也不返回任何值。

这里我们将OSTimeTickHook()写成空函数，代码见程序清单6.16。

程序清单6.16 函数OSTimeTickHook()

#if (OS_CPU_HOOKS_EN > 0) && (OS_TIME_TICK_HOOK_EN > 0)

void  OSTimeTickHook (void)

{
}

#endif
程序将OSTimeTickHook()定义为空函数。

6.4.2.8 系统初始化开始接口函数OSInitHookBegin()

进入OSInit()函数后，OSInitHookBegin()就会被立即调用。添加这个函数的原因在于，想把与OS有关的初始化代码也放在OSInit()函数中。这个函数使得用户可以将自己特定的代码也放在OSInit()函数中。这样代码会简洁明了


这里我们将OSInitHookBegin()写成空函数，代码见程序清单6.17。

程序清单6.17 函数OSInitHookBegin()

#if OS_CPU_HOOKS_EN > 0 && OS_VERSION > 203

void  OSInitHookBegin (void)

{

}

#endif
程序将OSInitHookBegin()定义为空函数。

6.4.2.9 系统初始化结束接口函数OSInitHookEnd()

OSInitHookEnd()函数与OSInitHookBegin()函数相似，只是它在OSInit()函数返回之前被调用。添加这个函数的原因与添加OSInitHookBegin()的原因是相同的。


这里我们将OSInitHookEnd()写成空函数，代码见程序清单6.18。

程序清单6.18 函数OSInitHookEnd()

#if OS_CPU_HOOKS_EN > 0 && OS_VERSION > 203

void  OSInitHookEnd (void)

{
}

#endif
程序将OSInitHookEnd()定义为空函数。

6.4.2.10 控制块初始化接口函数OSTCBInitHook()

OSTCBInit()函数在调用OSTaskCreateHook()之前，会先调用OSTCBInitHook()函数。这样做的原因在于，用户可以在OSTCBInitHook()函数中做一些与初始化任务控制块OS_TCB有关的处理；在OSTaskCreateHook()中做一些与初始化任务有关的处理。是否用OSTCBInitHook()或OSTaskCreateHook()函数，完全取决于用户。同OSTaskCreateHook()一样，OSTCBInitHook()会收到一个指向新添加任务的任务控制块的指针，而这个新添加任务的任务控制块绝大部分已经初始化完成，但是还没有链接到已经建立任务的链表中。


这里我们将OSTCBInitHook()写成空函数，代码见程序清单6.19。

程序清单6.19 函数OSTCBInitHook()

#if OS_CPU_HOOKS_EN > 0 && OS_VERSION > 203

void  OSTCBInitHook (OS_TCB *ptcb)

{

    ptcb = ptcb;         
/* 防止编译警告 */

}

#endif
程序将OSTCBInitHook()定义为空函数。

6.4.3 OS_CPU_A.S文件


µC/OS-II的移植范例要求用户编写4个简单的汇编函数：


OSStartHighRdy();


OSCtxSw();


OSIntCtxSw();


OSTickISR();

当然，因为ADS编译器支持插入行汇编代码，也可以将所有与处理器相关的代码放在OS_CPU_C.C文件中，而不必再有单独的汇编文件。

6.4.3.1 OSStartHighRdy()

µC/OS-II启动多任务环境的函数叫做OSStart()，用户在调用OSStart()函数之前，必须已经建立了一个或多个任务。OSStart()最终调用OSStartHighRdy()来使就绪态任务中优先级最高的任务开始运行，这个函数的示意性代码如下所示：

void OSStartHighRdy(void)

{


调用用户定义的OSTaskSwHook();


（1）


OSRunning = TRUE;





（2）


获取将要恢复运行任务的堆栈指针：

SP = OSTCBHighRdy->OSTCBStkPtr;

（3）


从新任务堆栈中恢复处理器的所有寄存器;
（4）


执行中断返回指令;





（5）

}

以上示意性代码对应的汇编代码见程序清单6.20。

程序清单6.20 OSStartHighRdy()函数代码

OSStartHighRdy

MSR
CPSR_c, #(NO_INT | SVC32_MODE)
;切换到管理模式，关闭IRQ和FIQ中断

  
LDR   
R0, =OSTaskSwHook     
;调用OSTaskSwHook();

（1）

  
MOV  
LR, PC

   
BX    
R0
  
LDR  
R4, =OSRunning         
;OSRunning = TRUE


（2）

   
MOV  
R5, #1

  
STRB  
R5, [R4]

LDR   
R4, =OSTCBHighRdy   
;获取最高优先级任务的TCB地址

   
LDR   
R4, [R4]                
;获取堆栈指针

   
LDR   
SP, [R4]                
;将SP指向新任务的堆栈

（3）

   
LDR 
R4,  [SP], #4           
;恢复新任务的CPSR


（4）

   
MSR   
SPSR_cxsf,R4

   
LDMFD
SP!, {R0-R12,LR,PC}^    
;恢复新任务的上下文，并返回
（5）


这部分代码是比较严格的按照µC/OS-II提供的示意性代码编写的，其中程序清单6.20（1）对应示意性代码的（1），程序清单6.20（2）对应示意性代码的（2），程序清单6.20（3）对应示意性代码的（3），程序清单6.20（4）和（5）对应示意性代码的（4）和（5）。

6.4.3.2 OSCtxSw()

在µC/OS-II中，任务切换时会调用宏（或是函数）OS_TASK_SW()，µC/OS-II建议OS_TASK_SW()通过某种途径最终调用函数OSCtxSw()。µC/OS-II提供的OSCtxSw()示意性代码如下所示：

void OSCtxSw(void)

{


保存处理器寄存器;








（1）


将当前任务的堆栈指针保存到当前任务的任务控制块中：



OSTCBCur->OSTCBStkPtr = 堆栈指针;



（2）


调用用户定义的OSTaskSwHook();





（3）


OSTCBCur = OSTCBHighRdy;






（4）


OSPrioCur = OSPrioHighRdy;






（5）


得到将要重新开始运行的任务的堆栈指针：





堆栈指针 = OSTCBHighRdy->OSTCBStkPtr;


（6）


从新任务的任务堆栈中恢复处理器的所有寄存器;


（7）


执行中断返回指令;








（8）

}

以上示意性代码对应的汇编代码见程序清单6.21。

程序清单6.21 OSCtxSw()函数

OSCtxSw

                              

;保存当前任务的上下文

STMFD 
SP!, {LR}          

;返回地址进栈




（1）

   
STMFD
SP!, {LR}



;R13进栈

   
STMFD 
SP!, {R0-R12}       

;R0-R12进栈

   
MRS   
R4, CPSR          

;当前CPSR进栈

   
TST   
LR, #1             

;是否是在Thumb状态下调用
   
ORRNE 
R4, R4, #0x20       

;如果是，则置CPSR的T位
  
STMFD 
SP!, {R4}

  
LDR  
R4, =OSTCBCur     

;OSTCBCur->OSTCBStkPtr = SP;
（2）

  
LDR   
R5, [R4]

 
STR   
SP, [R5]

 
LDR  
R0, =OSTaskSwHook 

;OSTaskSwHook();




（3）

 
MOV  
LR, PC

 
BX  
R0            

LDR  
R4, =OSPrioCur   

;OSPrioCur = OSPrioHighRdy;

（4）

LDR   
R5, =OSPrioHighRdy

LDRB  
R6, [R5]

STRB  
R6, [R4]

LDR   
R4, =OSTCBCur     

;OSTCBCur = OSTCBHighRdy;

（5）

LDR   
R6, =OSTCBHighRdy

LDR   
R6, [R6]

STR   
R6, [R4]

LDR   
SP, [R6]          
;SP = OSTCBHighRdy->OSTCBStkPtr;
（6）

                              

;恢复新任务的上下文

LDMFD
SP!, {R4}           

;恢复新任务的CPSR



（7）

MSR   
SPSR_cxsf, R4
LDMFD
SP!, {R0-R12,LR,PC}^ 
;恢复新任务的所有寄存器，并返回
（8）

这部分代码是比较严格的按照µC/OS-II提供的示意性代码编写的，其中程序清单6.21（1）对应示意性代码的（1），程序清单6.21（2）对应示意性代码的（2），程序清单6.21（3）对应示意性代码的（3），程序清单6.21（4）对应示意性代码的（4），程序清单6.21（5）对应示意性代码的（5），程序清单6.21（6）对应示意性代码的（6），程序清单6.21（7）和（8）对应示意性代码的（7）和（8）。

6.4.3.3 OSIntCtxSw()

在µC/OS-II中，中断退出函数通过调用OSIntCtxSw()函数，在ISR中执行切换功能。µC/OS-II提供的OSIntCtxSw()示意性代码如下所示：

void OSIntCtxSw (void)

{


调用用户定义的OSTaskSwHook();




（1）


OSTCBCur = OSTCBHighRdy;





（2）


OSPrioCur = OSPrioHighRdy;





（3）


得到将要重新开始运行的任务的堆栈指针：



堆栈指针 = OSTCBHighRdy->OSTCBStkPtr;

（4）


从新任务的任务堆栈中恢复处理器的所有寄存器;

（5）


执行中断返回指令;







（6）

}


比较OSIntCtxSw()函数和OSCtxSw()函数的示意性代码可以看出，OSIntCtxSw()函数中的绝大多数代码同OSCtxSw()函数是一样的。区别只是，因为ISR已经保存了CPU的寄存器，所以在OSIntCtxSw()函数中就不需要再保存CPU的寄存器了。


OSIntCtxSw()函数的汇编代码见程序清单6.22。

程序清单6.22 OSIntCtxSw()函数

OSIntCtxSw

  
LDR 
R0, =OSTaskSwHook 

;OSTaskSwHook();




（1）

MOV  
LR, PC

BX    
R0            

LDR  
R4, =OSPrioCur         
;OSPrioCur = OSPrioHighRdy;

（2）

LDR  
R5, =OSPrioHighRdy

LDRB 
R6, [R5]

STRB 
R6, [R4]

LDR  
R4, =OSTCBCur        
;OSTCBCur = OSTCBHighRdy;

（3）

LDR  
R6, =OSTCBHighRdy

LDR  
R6, [R6]

STR   
R6, [R4]

LDR  
SP, [R6]            
;SP = OSTCBHighRdy->OSTCBStkPtr;
（4）

                                  
;恢复新任务的上下文
LDMFD
SP!, {R4}               
;恢复新任务的CPSR



（5）

MSR   
SPSR_cxsf, R4
LDMFD
SP!, {R0-R12,LR,PC}^    
;恢复新任务的所有寄存器，并返回
（6）

这部分代码是比较严格的按照µC/OS-II提供的示意性代码编写的，其中程序清单6.22（1）对应示意性代码的（1），程序清单6.22（2）对应示意性代码的（2），程序清单6.22（3）对应示意性代码的（3），程序清单6.22（4）对应示意性代码的（4），程序清单6.22（5）和（6）对应示意性代码的（5）和（6）。

6.4.3.4 中断服务程序和OSTickISR()
在µC/OS-II中，中断服务子程序编写的要求如下所示：

中断服务子程序：


保存全部CPU寄存器;


调用OSIntEnter()或OSIntNesting直接加1;


if(OSIntNesting == 1){



OSTCBCur->OSTCBStkPtr = SP;


}


清除中断源;


重新开中断;


执行用户代码做中断服务;


调用OSIntExit();


恢复所有CPU寄存器;


执行中断返回指令;

对应上面中断服务子程序编写要求，IRQ和FIQ中断服务程序的代码见程序清单6.23和程序清单6.24。
程序清单6.23 IRQ中断服务程序
OS_CPU_IRQ_ISR

  
STMFD   SP!, {R1-R3} 
;将工作寄存器压入IRQ堆栈

MOV     R1, SP  

;保存IRQ堆栈指针

ADD     SP, SP,#12  

;调整IRQ堆栈指针
SUB     R2, LR,#4   
;调整任务的返回地址

MRS     R3, SPSR   
;将SPSR（即被中断的任务的CPSR）复制到R3中

MSR     CPSR_c, #(NO_INT | SVC32_MODE) 
;切换到SVC模式

                                      
;将任务的上下文保存到任务堆栈中

STMFD   SP!, {R2}              

;任务的返回地址入栈
STMFD   SP!, {LR}           

;任务的LR入栈
STMFD   SP!, {R4-R12}        

;任务的R12-R4入栈

LDMFD   R1!, {R4-R6}  ;将任务的R1-R3 从IRQ堆栈转移到SVC堆栈

STMFD   SP!, {R4-R6}

STMFD   SP!, {R0}    
;将任务的R0压入任务的堆栈

STMFD   SP!, {R3}    
;将任务的CPSR（即IRQ的SPSR）压入任务的堆栈
                     




;处理中断嵌套指针

LDR     R0, =OSIntNesting          
;OSIntNesting++;

LDRB    R1, [R0]

ADD     R1, R1,#1

STRB    R1, [R0]

CMP     R1, #1             

;if (OSIntNesting == 1) {

BNE     OS_CPU_IRQ_ISR_1

LDR     R4, =OSTCBCur



;OSTCBCur->OSTCBStkPtr = SP

LDR     R5, [R4]

STR     SP, [R5]            

;}

OS_CPU_IRQ_ISR_1

  
MSR CPSR_c, #(NO_INT | IRQ32_MODE)


;切换到IRQ模式

;（使用IRQ 堆栈处理中断）

EIC_base_addr         EQU    0xFFFFF800 
;EIC起始地址

CICR_off_addr         EQU    0x04      

;当前中断通道寄存器

IVR_off_addr          EQU    0x18      

;中断向量寄存器

IPR_off_addr          EQU    0x40      

;中断标志寄存器

LDR     R0, =(EIC_base_addr + IVR_off_addr)

MOV     LR, PC

BX      R0                    
;跳转到相应IRQ处理程序


（1）

                                      
;清除EIC的中断标志（使用相应的IPRx）

LDR     R0, =EIC_base_addr

LDR     R2, [R0, #CICR_off_addr]    


;获取IRQ通道号

MOV     R3, #1

MOV     R3, R3, LSL r2

STR     R3, [R0, #IPR_off_addr]       
 

;清除对应的IPR位

MSR     CPSR_c, #(NO_INT | SVC32_MODE) 
;切换到管理模式

LDR     R0, =OSIntExit                 

;OSIntExit();

MOV     LR, PC

BX      R0            

                                           
;恢复新任务的上下文
LDMFD   SP!, {R4}                     
;恢复新任务的CPSR

MSR     SPSR_cxsf, R4

LDMFD   SP!, {R0-R12,LR,PC}^         
;恢复新任务寄存器，并返回

程序清单6.24 FIQ中断服务程序

OS_CPU_FIQ_ISR
STMFD   SP!, {R1-R3}                 
;将工作寄存器压入FIQ堆栈 

MOV     R1, SP                       
;保存FIQ堆栈指针

ADD     SP, SP,#12                     
;调整FIQ堆栈指针

SUB     R2, LR,#4                      
;调整任务的返回地址

MRS     R3, SPSR    
;将SPSR（即被中断的任务的CPSR）复制到R3中
MSR     CPSR_c, #(NO_INT | SVC32_MODE) 
;切换到SVC模式
                                  
;将任务的上下文保存到任务堆栈中
STMFD   SP!, {R2}             
;任务的返回地址入栈
STMFD   SP!, {LR}            
;任务的LR入栈
STMFD   SP!, {R4-R12}        
;任务的R12-R4入栈
LDMFD   R1!, {R4-R6}  
;将任务的R1-R3 从FIQ堆栈转移到SVC堆栈
STMFD   SP!, {R4-R6}

STMFD   SP!, {R0}    
;将任务的R0压入任务的堆栈
STMFD   SP!, {R3}   
;将任务的CPSR（即FIQ的SPSR）压入任务的堆栈
                                    
;处理中断嵌套指针
LDR     R0, =OSIntNesting          
;OSIntNesting++;

LDRB    R1, [R0]

ADD     R1, R1,#1

STRB    R1, [R0]

CMP     R1, #1                    
;if (OSIntNesting == 1) {

BNE     OS_CPU_FIQ_ISR_1

LDR     R4, =OSTCBCur       


;OSTCBCur->OSTCBStkPtr = SP

LDR     R5, [R4]

STR     SP, [R5]                   
;}

OS_CPU_FIQ_ISR_1

MSR     CPSR_c, #(NO_INT | FIQ32_MODE) 
;切换到FIQ模式

;（使用FIQ 堆栈处理中断）
LDR     R0, =FIQ_Handler               

;调用FIQ_Handler()
MOV     LR, PC

BX      R0            

MSR     CPSR_c, #(NO_INT | SVC32_MODE) 
;切换到管理模式
LDR     R0, =OSIntExit                 

;OSIntExit();

MOV     LR, PC

BX      R0            

                                               
;恢复新任务的上下文
LDMFD   SP!, {R4}                      
;恢复新任务的CPSR

MSR     SPSR_cxsf, R4

LDMFD   SP!, {R0-R12,LR,PC}^           
;恢复新任务寄存器，并返回

至于时钟节拍中断服务程序的编写，除了在调用的用户处理程序（程序清单6.23 （1）中必须调用函数OSTimeTick()外，没有其它不同。这里采用RTC产生µC/OS-II所需的时钟节拍。代码见程序清单6.24。

程序清单6.24 时钟节拍中断服务程序

void RTC_IRQHandler(void)



// OSTickISR()

{


if(RTC_FlagStatus(RTC_SIR) == SET)


{



OSTimeTick();



// 调用OSTimeTick()


RTC_FlagClear(RTC_SIR);



RTC_FlagClear(RTC_GIR);


}

}
6.5 µC/OS-II在STR71x评估板的使用举例

根据以上步骤，在STR71x评估开发板（DV710B）上移植好µC/OS-II后，下面列举二个使用µC/OS-II的简单程序，以帮助读者理解、应用。
6.5.1 示例1


首先建立一个名为TaskCount的任务，该任务负责刷新当前时间，并将CPU的使用率显示在LCD上。TaskCount再建立3个任务：Task1，Task2，Task3；每个任务负责闪一个灯，分别为LED1，LED2，LED3。各个任务和主程序的代码见程序清单6.25。
程序清单6.25 应用示例1

#define STKSIZE 128

OS_STK Stack1[STKSIZE], Stack2[STKSIZE], Stack3[STKSIZE]; // 任务堆栈

OS_STK StackCount[STKSIZE];

void Task1(void *arg)



// 任务1：负责闪LED1

{


while(1)


{



LED_Set(0, LED_TOGGLE);



OSTimeDly(128);


}

}

void Task2(void *arg)



// 任务2：负责闪LED2

{


while(1)


{



LED_Set(1, LED_TOGGLE);



OSTimeDly(256);


}

}

void Task3(void *arg)



// 任务3：负责闪LED3

{


while(1)


{



LED_Set(2, LED_TOGGLE);



OSTimeDly(512);


}

}
void TaskCount(void *arg)



// 任务TaskCount
{

OSTaskCreate(Task1, 0, Stack1 + (STKSIZE - 1), 1);

// 建立3个任务


OSTaskCreate(Task2, 0, Stack2 + (STKSIZE - 1), 2);


OSTaskCreate(Task3, 0, Stack3 + (STKSIZE - 1), 3);


LCD_Goto(0, 0);


LCD_Puts("00:00:00:00");


LCD_Goto(1, 0);


LCD_Puts("CPU Usage:");


while(1)


{



OSTimeDly(1);



if(++ms >= OS_TICKS_PER_SEC)



{




ms = 0;




if(++ss >= 60)




{





ss = 0;





if(++mm >= 60)





{






mm = 0;






++hh;






LCD_Goto(0, 0);






LCD_Printf("%02u", hh);

// 显示小时





}





LCD_Goto(0, 3);





LCD_Printf("%02u", mm);


// 显示分钟




}




LCD_Goto(0, 6);




LCD_Printf("%02u", ss);



// 显示秒


}



LCD_Goto(0, 9);



LCD_Printf("%02u", ms * 100 / OS_TICKS_PER_SEC);
// 显示毫秒



LCD_Goto(1, 10);



LCD_Printf("%3d%%", OSCPUUsage);



// 显示CPU使用率


}

}
void Main(void)

{

OSInit();

// 初始化µC/OS-II 


OSTaskCreate(TaskCount, 0, StackCount + (STKSIZE - 1), 4);
// 建立任务TaskCount

OSStart();
// 启动µC/OS-II
}
程序首先调用OSInit()初始化µC/OS-II，然后建立一个任务TaskCount，最后启动µC/OS-II。由于这时只有任务TaskCount处于就绪态，启动内核后任务TaskCount开始执行。任务TaskCount建立3个任务：Task1，Task2，Task3。
6.5.2 示例2

下面举一个使用消息邮箱的例子。

首先建立任务AppTaskStart，该任务负责闪LED2。任务AppTaskStart调用AppTaskCreate函数建立两个任务：Task1，Task2。Task1在键1按下时将LED1关闭，在键2按下时发送一条消息；Task2等待消息，如果有消息，则点亮LED1。各个任务和主程序的代码见程序清单6.26。代码有的地方使用伪代码，以便于理解。
程序清单6.26 应用示例2

#define  APP_TASK_START_PRIO          5

// 任务优先级
#define  APP_TASK_START_STK_SIZE    100

// 任务堆栈长度
#define  STKSIZE  100
OS_STK   AppTaskStartStk[APP_TASK_START_STK_SIZE];

// 任务堆栈

OS_STK   Stack1[STKSIZE], Stack2[STKSIZE];
static  void  Task1(void *p_arg)

// 任务Task1

{
while(1)


{



读键1;



if(键1按下)







将LED1关闭;



读键2;



if(键2按下)




发送消息OSMboxPost;

// 发送消息



OSTimeDlyHMSM(0, 0, 0, 100);


}
}

static  void  Task2 (void *p_arg)


// 任务Task2 

{

LCD_init(); 
// 初始化LCD

while(1)


{


等待消息OSMboxPend;


// 等待消息，如果有消息，将LED1点亮


将LED1点亮;

}
}

static  void  AppTaskCreate (void)

// 这里只建立AppTaskUserIF任务

{

OSTaskCreate(Task1, 0, Stack1 + (STKSIZE - 1), 1);

// 建立3个任务


OSTaskCreate(Task2, 0, Stack2 + (STKSIZE - 1), 2);
}

static  void  AppTaskStart (void *p_arg)

{
LED_Off(0);      
// 关闭所以LED灯

AppTaskCreate();     
// 建立应用程序任务

while (TRUE) {  

/*闪LED2*/

  
OSTimeDlyHMSM(0, 0, 0, 100);

}

}
void  main (void)

{
    OSInit();    
// 初始化µC/OS-II
    OSTaskCreateExt(AppTaskStart,   // 建立第一个任务

                    (void *)0,

                    (OS_STK *)&AppTaskStartStk[APP_TASK_START_STK_SIZE - 1],

                    APP_TASK_START_PRIO,

                    APP_TASK_START_PRIO,

                    (OS_STK *)&AppTaskStartStk[0],

                    APP_TASK_START_STK_SIZE,

                    (void *)0,

                    OS_TASK_OPT_STK_CHK | OS_TASK_OPT_STK_CLR);
    OSStart();  
// 启动µC/OS-II
}
程序首先调用OSInit()初始化µC/OS-II，然后建立一个任务AppTaskStart，最后启动µC/OS-II。由于这时只有任务AppTaskStart处于就绪态，启动内核后任务AppTaskStart开始执行。任务AppTaskStart建立两个任务：Task1，Task2。Task1实现的功能为：当键1按下时，将LED1关闭，当键2按下时，发送消息；Task2实现的功能为：阻塞等待消息，如果有消息，就将LED1打开。这个程序很好地演示了利用消息邮箱进行任务间通信的过程。

